Motorola Semiconductor Application Note

AN1774

Interfacing the MC68HC912B32 to an LCD Module

By Mark Glenewinkel Field Applications Engineering Austin, Texas

Introduction

More and more applications are requiring liquid crystal displays (LCD) to communicate effectively to the outside world. This application note describes the hardware and software interface needed to display information from the MC68HC912B32 (B32).

Some LCD suppliers provide only the LCD glass so that the waveforms needed to directly drive the LCD segments have to be generated by the microcontroller (MCU) or microprocessor (MPU). Other LCD suppliers provide an LCD module, which has all LCD glass and segment drivers provided in one small packaged circuit board.

This application note uses an LCD module from Optrex Corporation, part number DMC16207 (207). It utilizes a Hitachi LCD driver, HD44780, to provide the LCD segment waveforms and a simple parallel port interface that easily interfaces to an MCU or MPU bus.

Circuitry and example code are given to also demonstrate the ability of providing pre-defined messages from memory to the display. The code can be modified easily to take serial peripheral interface (SPI) and serial communication interface (SCI) data and display it on the LCD module.



© Motorola, Inc., 1999 AN1774 Rev. 1.0

LCD Module Hardware Interface

Optrex has many LCD module configurations that have varying display lines and display line character lengths. The 207 module has a 2-line, 16-character per line display. Each character is displayed using a 5 x 7 pixel font matrix. The 207 module has a character generator ROM capable of displaying ASCII characters.

The parallel interface bus can work with either 4-bit or 8-bit buses. Once data is presented on the bus, it is latched by clocking the E pin on the device. Depending on the RS pin, the data will be used as an instruction or an ASCII character.

Pin Descriptions

Table 1 describes the interface pins found on the 207 module.

Table 1. 207 Module Pinout

Pin Number	Signal	I/O	Function
1	V _{SS}	Power	GND (ground)
2	V _{CC}	Power	4.5 volts to 5.5 volts
3	V _{EE}	Power	LCD drive voltage
4	RS	I	Selects registers 0: Instruction register (for write), address counter (for read) 1: Data register (for write and read)
5	R/W	I	Selects read or write 0: Write 1: Read
6	E	I	Starts data read/write on falling edge
14–11	DB7-DB4	I/O	Four high order bidirectional 3-state data bus pins. Used for data transfer and receive between the MCU and the 207. DB7 can be used as a busy flag.
10–7	DB3-DB0	I/O	Four low order, bidirectional, 3-state data bus pins. Used for data transfer and receive between the MCU and the 207. These pins are not used during 4-bit operation.

AN1774

Bus Timing

Table 2. Bus Timing Electricals

Spec	Symbol	Min	Тур	Max	Unit
Enable cycle time	t _{CYCLE}	500	_	_	ns
Enable pulse width (high level)	PW _{EH}	230	_	_	ns
Enable rise and decay time	t _{Er} , t _{Ef}	_	_	20	ns
Address setup time, RS, R/W, E	t _{AS}	40	_	_	ns
Data delay time	t _{DDR}	_	_	160	ns
Data setup time	t _{DSW}	80	_	_	ns
Data hold time (write)	t _H	10	_	_	ns
Data hold time (read)	t _{DHR}	5	_		ns
Address hold time	t _{AH}	10	_	_	ns




Figure 1. Write Timing Operation




Figure 2. Read Timing Operation

AN1774

Bus Interface

Figure 3 and Figure 4 show examples of 8-bit and 4-bit timing sequences, respectively.

NOTE:

A BF (busy flag) check is not needed if the maximum instruction execution time is respected before sending another instruction.




Figure 3. 8-Bit Bus Timing Sequence




Figure 4. 4-Bit Bus Timing Sequence

For 4-bit interface data, only four bus lines (DB7–DB4) are used for transfer.

Bus lines DB3-DB0 are disabled.

AN1774

The data transfer is completed after the 4-bit data has been transferred twice.

The four high order bits are transferred first (DB7–DB4), and then the low order bits are transferred (DB3–DB0).

LCD Module Software Interface

LCD Instruction Commands

The 207 module has many different configurations that can be implemented easily by sending the correct function command to the device. These commands are listed in **Table 3** followed by an explanation of each function they execute.

Table 3. 207 Module Instruction Code

Instruction	RS	R₩	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0	Execution Time (max)
Clear display	0	0	0	0	0	0	0	0	0	1	1.64 ms
Return cursor home	0	0	0	0	0	0	0	0	1	х	1.64 ms
Entry mode set	0	0	0	0	0	0	0	1	I/D	S	40 μs
Display on/off control	0	0	0	0	0	0	1	D	С	В	40 μs
Cursor or display shift	0	0	0	0	0	1	S/C	R/L	х	х	40 μs
Function set	0	0	0	0	1	DL	N	F	х	х	40 μs
Set CGRAM address	0	0	0	1	A _{CG}	40 μs					
Set DDRAM address	0	0	1	A _{DD}	40 μs						
Read busy flag and address	0	1	BF	A _C	0 μs						
Write data to CG or DDRAM	1	0	D7	D6	D5	D4	D3	D2	D1	D0	40 μs
Read data from CG or DDRAM	1	1	D7	D6	D5	D4	D3	D2	D1	D0	40 μs

DDRAM: Display data RAM CGRAM: Character generator RAM

A_{CG}: CGRAM address

A_{DD}: DDRAM address; corresponds to cursor address

A_C: Address counter used for both DDRAM and CGRAM addresses

AN1774

Clear Display

Clear display writes space code \$20 into all DDRAM addresses. It then sets DDRAM address 0 into the address counter and returns the display to its original status if it was shifted. In other words, the display disappears and the cursor or blinking goes to the left edge of the first line of the display. I/D of entry mode is set to 1 (increment mode). S of entry mode is left unchanged.

Return Cursor Home

Return cursor home sets the DDRAM address 0 into the address counter and returns the display to its original status if it was shifted. The DDRAM contents do not change.

The cursor or blinking goes to the left edge of the first line of the display.

Entry Mode Set

I/D — Increments (I/D = 1) or decrements (I/D = 0) the DDRAM address by 1 when a character code is written into or read from DDRAM. The cursor or blinking moves to the right when incremented by 1 and to the left when decremented by 1. The same applies to writing and reading of CGRAM.

S — Shifts the entire display either to the right (I/D = 0) or to the left (I/D = 1) when S is 1. The display does not shift if S is 0. If S is 1, it will seem as if the cursor does not move but the display does. The display does not shift when reading from DDRAM. Also, writing into or reading out from CGRAM does not shift the display.

Display On/Off Control

D — The display is on when D = 1 and is off when D = 0. When off, the display data remains in DDRAM, but it can be displayed instantly by setting D = 1.

 ${f C}$ — The cursor is displayed when ${f C}=1$ and not displayed when ${f C}=0$. Even if the cursor disappears, the function of I/D or other specifications will not change during display data write. The cursor is displayed using five dots in the eighth line of the 5 x 8 dot character.

B — The character indicated by the cursor blinks when B = 1. The blinking is displayed as switching between all blank dots and displayed characters at a speed of 409.6-ms intervals when f_{OSC} (HD44780 operating frequency) is 250 kHz. The cursor and blinking can be set to display simultaneously. (The blinking frequency changes according to f_{OSC} . For example, when f_{OSC} , is 270 kHz, $409.6 \times (250/270) = 379.2 \text{ ms.}$)

AN1774

Cursor or Display Shift Cursor or display shift shifts the cursor position or display to the right or left without writing or reading display data. (See **Table 4**.) This function is used to correct or search the display. In a 2-line display, the cursor moves to the second line when it passes the 40th digit of the first line. The first and second line displays will shift at the same time.

When the displayed data is shifted repeatedly, each line moves only horizontally. The second line display does not shift into the first line position.

The address counter (A_C) contents will not change if the only action performed is a display shift.

 S/C
 R/L
 Description

 0
 0
 Shifts the cursor position to the left; A_C is decremented by 1

 0
 1
 Shifts the cursor position to the right; A_C is incremented by 1

 1
 0
 Shifts the entire display to the left; the cursor follows the display shift

 1
 1
 Shifts the entire display to the right; the cursor follows the display shift

Table 4. Cursor and Display Shift Combination

Function Set

DL — Sets the interface data length. Data is sent or received in 8-bit lengths (DB7 to DB0) when DL = 1 and in 4-bit lengths (DB7 to DB4) when DL = 0. When 4-bit length is selected, data must be sent or received twice.

N — Sets the number of display lines

F — Sets the character font

NOTE:

Perform the function set instruction at the beginning of the program before executing any instructions (except for the read busy flag and address instruction). From this point, the function set instruction cannot be executed unless the interface data length is changed.

AN1774

Set CGRAM Address Set CGRAM address sets the CGRAM binary address $A_{CG}5-A_{CG}0$ into the address counter. Data is written to or read from the MCU for CGRAM.

Set DDRAM Address Set DDRAM address sets the DDRAM binary address $A_{DD}6-A_{DD}0$ into the address counter. Data is written to or read from the MCU for DDRAM.

Read Busy Flag and Address Read busy flag and address reads the busy flag (BF) indicating that the system is now internally operating on a previously received instruction. If BF = 1, the internal operation is in progress. The next instruction will not be accepted until BF is reset to 0. Check the BF status before the next write operation. At the same time, the value of the address counter in binary (A_C6-A_C0) is read out. This address counter is used by both CGRAM and DDRAM addresses, and its value is determined by the previous instruction. The address contents are the same as for instructions set CGRAM address and set DDRAM address.

Write Data to CGRAM or DDRAM Write data to CGRAM or DDRAM writes 8-bit data to CGRAM or DDRAM. To write into CGRAM or DDRAM is determined by the previous specification of the CGRAM or DDRAM address setting. After a write, the address is incremented or decremented automatically by 1 according to the entry mode. The entry mode also determines the display shift.

Read Data from CGRAM or DDRAM Read data from CGRAM or DDRAM reads 8-bit data from CGRAM or DDRAM. The previous designation determines whether CGRAM or DDRAM is to be read. Before entering this read instruction, either CGRAM or DDRAM address set instruction must be executed. If not executed, the first read data will be invalid. When serially executing read instructions, the next address data normally is read from the second read. The address set instructions need not be executed just before this read instruction when shifting the cursor by the cursor shift instruction (when reading out of DDRAM).

AN1774

The operation of the cursor shift instruction is the same as the set DDRAM address instruction. After a read, the entry mode automatically increases or decreases the address by 1. However, the display shift is not executed regardless of the entry mode.

Address Map

Table 5 shows the address map for the HD44780. The character positions of the LCD module are shown in the first row of the table with the addresses shown beneath them. The 207 uses only the first 16 addresses.

NOTE:

The addresses are seven bits wide and when writing to the DDRAM, the MSB (bit 7) is always a 1. Therefore, to write to address \$02, the 8-bit data sent to the 207 will be \$82 or binary 10000010%.

Understand that when the display is shifted, the whole address map is used. In other words, when a shift right is executed, the character at address \$27 is moved to position 1 of the first line of the display.

Bit Bit Bit Bit Bit Bit Bit Bit ... 1 2 3 4 5 16 39 40 \$00 \$01 \$02 \$03 \$04 \$26 \$27 \$0F \$40 \$41 \$42 \$43 \$44 \$4F \$66 \$67

Table 5. LCD Address Map

Initialization Routines

To ensure proper initialization of the 207 module, a sequence of instruction codes must be executed. These instructions set the data bus width, font type, and number of display lines. In addition, the LCD is cleared, and the entry mode for data is set.

Figure 5 shows the power-on reset initialization for an 8-bit data bus, while **Figure 6** shows the power-on reset initialization for a 4-bit data bus.

AN1774




Figure 5. Power-On Reset 8-Bit Initialization

AN1774




Figure 6. Power-On Reset 4-Bit Initialization

AN1774

MC68HC912B32 Hardware Interface

The B32 is a 16-bit MCU device with standard on-chip peripherals including:

- 32 Kbytes of FLASH EEPROM
- 1 Kbyte of RAM
- 768 bytes of EEPROM
- Asychronous serial communications interface (SCI)
- Serial peripheral interface (SPI)
- 8-channel, 16-bit timer
- 8-channel, 8-bit analog-to-digital converter (ADC)
- 4-channel pulse-width modulator (PWM)
- J1850-compatible byte data link communications module (BDLC)

The B32 has a maximum of 63 I/O (input/output) pins in single-chip mode. These I/O pins share functionality with the on-chip peripheral modules. Rarely will a system have all of these I/O pins available. The LCD module works in either an 8-bit or 4-bit data bus. The data bus size should be defined from the I/O, peripheral, and code space usage of the application. Three I/O pins are also needed for bus control.

The schematic used for testing the B32-to-207 interface on the MC68HC912B32 evaluation board is shown in **Figure 7**. The test circuit was designed to use either a 4-bit or 8-bit databus. Although the R/W pin on the 207 is connected to the B32, it may be grounded if only writes to the LCD are executed. Since we cannot check the BF flag, the delay times stated in **Table 3** must be observed.

Although these routines were tested on an MC68HC912B32 device, any HC12 device with enough memory and I/O can execute these routines. A simple change in the memory map should allow the code to be ported to other HC12s.

AN1774




Figure 7. B32-to-207 Interface Test Circuit

MC68HC912B32 Software Interface

The software written to demonstrate the MC68HC912B32-to-LCD module interface is shown in sections titled **Flowcharts**, **4-Bit Bus Code**, and **8-Bit Bus Code**.

The flowchart roughly sketches out the routines.

The code was written to take pre-defined messages in ROM and easily display them by calling a subroutine. If the B32 is receiving messages from the SPI or SCI, put the ASCII data in a temporary RAM buffer and change the message routines to start reading ASCII characters from the start of the buffer.

AN1774

Development Tools

The interface was created and tested using these development tools:

- M68HC12B32EVB Motorola's MC68HC912B32 evaluation board
- WIN IDE— P&E Microcomputer Systems integrated development environment, version 1.02
- CASM12W P&E Microcomputer Systems HC12 assembler, version 3.08
- ICD12W P&E Microcomputer Systems HC12 in-circuit debugger, version 1.04 build B

References

MC68HC912B32 Technical Summary, Motorola document order number MC68HC912B32TS/D, 1997.

M68HC12 CPU12 Reference Manual, Motorola document order number CPU12RM/AD, 1997.

DMC-16207 Digikey #73-1025-ND.

1997 Optrex LCD Databook Digikey #73-1001-ND.

Motorola's HC12 website:

http://www.mcu.motsps.com/hc12/index.html




Figure 8. Main Flowchart

AN1774




Figure 9. LCD_Write Subroutine Flowchart




Figure 10. LCD_ADDR Subroutine Flowchart

16 MOTOROLA

AN1774




Figure 11. Message Subroutine Flowchart

AN1774

8-Bit Bus Code

```
* File name: H12 LCD8.ASM
* Example Code for LCD Module (DMC16207) using 8-bit bus
   interfacing with the MC68HC912B32
* Ver: 1.0
* Date: September 6, 1998
* Author: Mark Glenewinkel
       Motorola Field Applications
* Assembler: P&E CASM12W ver 3.08
*For code explanation and flowcharts, please consult Motorola Application Note
    "Interfacing the MC68HC912B32 to an LCD Module" Literature # AN1774/D
* Note: Code originates in RAM instead of FLASH
*************************
*** SYSTEM DEFINITIONS AND EQUATES *******************************
*** Internal Register Definitions
          EQU
                 $00
                                          ;LCD data bus
PORTB
          EOU
                  $01
                                          ;LCD control signals
DDRA
          EQU
                  $02
                                          ;data direction for PortA
DDRB
           EQU
                  $03
                                          ;data direction for PortB
*** Application Specific Definitions
LCD DATA
        EQU
               $00
                                          ; PORTA
LCD_CTRL
           EOU
                  $01
                                          ; PORTB
           EQU
                 1T
                                          ; PORTB, bit 0
F.
                                          ; PORTB, bit 2
RW
           EQU
                 4T
                                          ; PORTB, bit 1
RS
           EQU
                  2Т
*** Memory Definitions
                 $0800
RAM START
          EOU
                                          ;start of RAM mem
RAM VAR
           EQU
                 $0BF0
                                          ;start of RAM variables
MSG_STORAGE
          EQU
                  $0B00
                                          ;start of message block
*** Vectors
RESET
           EQU
                  $FFFE
                                          ; vector for reset
ORG
                  RAM_VAR
TTME:
           DB
                                          ;used for delay time
ORG
                 RAM START
                                          ;start at begining of RAM
*** Initialize the Stack Pointer
           lds #$0BFF
                                          ;init SP, top or RAM
```

*** Wait for 15ms movb #150T,TIME set delay time sub for 0.1ms delay *** Send Init Command movb #\$38,LCD_DATA best LCD_CTRI,E belr LCD_CTRI,E *** Wait for 4.1ms movb #41T,TIME set delay time sub for 0.1ms delay *** Send Init Command movb #\$38,LCD_DATA belr CLD_CTRI,E *** Wait for 4.1ms movb #41T,TIME set delay time sub for 0.1ms delay *** Send Init Command movb #538,LCD_DATA belr LCD_CTRI,E *** Wait for 100 us movb #1T.TIME set delay time sub for 0.1ms delay *** Send Init Command sub for VAR_DELAY sub for 0.1ms delay *** Send Init Command sub for VAR_DELAY sub for 0.1ms delay *** Send Init Command sub for VAR_DELAY sub for 0.1ms delay *** Send Init Command sub for VAR_DELAY sub for 0.1ms delay *** Send Function Set Command *** Send Function Set Command *** Send Function Set Command *** Send Display Ctrl Command *** Send Display Ctrl Command *** Gisplay on, cursor off, no blinking ldaa #\$38 sub for 0.1ms delay *** Send Clear Display Command *** Clear display, cursor addr=0 ldaa #\$01 sub for 0.1ms delay *** Send Clear Display Cursor addr=0 ldaa #\$01 sub for 0.1ms delay *** Send Clear Display Cursor addr=0 ldaa #\$01 sub for 0.1ms delay *** Send Clear Display Cursor addr=0 sub for 0.1ms delay *** Send Entry Mode Command *** Send Entry Mode Command *** increment, no display shift ldaa #\$06 sub for 0.1ms delay *** Send Entry Mode Command *** MESSAGE1 send Message1 send Message2 DUMMY bra DUMMY sub with example	*** Intialize START	Ports clr clr movb movb	LCD_CTRL LCD_DATA #\$FF,DDRA #\$FF,DDRB	<pre>;clear LCD_CTRL ;clear LCD_DATA ;PortA output ;PortB output</pre>
### Send Init Command ### Sond Init Command ### Wait for 4.1ms ### word #41T.TIME ### bet LCD_CTRL,E ### word #41T.TIME ### bet LCD_CTRL,E ### word #41T.TIME ### bet LCD_CTRL,E ### word #41T.TIME ### jsr VAR_DELAY ### Send Init Command ### movb #\$38,LCD_DATA ### bet LCD_CTRL,E ### wait for 0.1ms delay ### Send Init Command ### word #1T.TIME ### wait for 100 us ### word #1T.TIME ### jsr VAR_DELAY ### Send Init Command ### ldaa #\$38 ### jsr LCD_WRITE ### send Function Set Command ### \$8 bit bus, 2 rows, 5x7 dots ### daa #\$38 ### jsr LCD_WRITE ### Send Display Ctrl Command ### display on, cursor off, no blinking ### daa #\$0C ### jsr LCD_WRITE ### clear Display Command ### send Clear Display Command ### clear Display Command #### clear Display Command ### clear Display Command #### clear Display Command ##### clear Display Command #### clear Display Command ##### clear Display Command ####################################			D	
movb #\$38,LCD_DATA /clock in data beet LCD_CTRL,E bclr LCD_CTRL,E /clock in data	*** Wait for	movb		
#41T.TIME	*** Send Init	movb bset	LCD_CTRL,E	
*** Send Init Command movb #\$38,LCD_DATA bset LCD_CTRL,E bclr LCD_CTRL,E bclr LCD_CTRL,E *** Wait for 100 us movb #1T,TIME jsr VAR_DELAY *** Send Init Command ldaa #\$38 jsr LCD_WRITE *** Send Function Set Command *** 8 bit bus, 2 rows, 5x7 dots ldaa #\$38 jsr LCD_WRITE *** Send Display Ctrl Command *** display on, cursor off, no blinking ldaa #\$00 jsr LCD_WRITE *** Send Clear Display Command *** clear display, cursor addr=0 ldaa #\$01 jsr LCD_WRITE *** Send Entry Mode Command *** clear display command *** clear display cursor addr=0 ldaa #\$01 jsr LCD_WRITE movb #16T,TIME jsr VAR_DELAY *** Send Entry Mode Command *** Send Messagel jsr MESSAGE1 jsr MESSAGE1 jsr MESSAGE1 jsr MESSAGE2 *** Messages have address and content predefined jsr MESSAGE1 jsr MESSAGE2 *** Messages have address and content predefined jsr MESSAGE2 *** Messages have address and content predefined jsr MESSAGE2 *** Messages have address and content predefined jsr MESSAGE2 *** Messages have address and content predefined jsr MESSAGE2 *** Messages have address and content predefined jsr MESSAGE2 *** Messages have address and content predefined jsr MESSAGE2 *** Messages have address and content predefined jsr MESSAGE2 *** Messages have address and content predefined jsr MESSAGE2 *** Messages have address and content predefined jsr MESSAGE2 *** Messages have address and content predefined jsr MESSAGE2 *** Messages have address and content predefined jsr MESSAGE2	*** Wait for	movb		
movb #\$38,LCD_DATA best LCD_CTRL,E bolr LCD_CTRL,E bolr LCD_CTRL,E bolr LCD_CTRL,E *** Wait for 100 us movb #1T,TIME sub for 0.1ms delay *** Send Init Command ldaa #\$38			_	-
movb #1T,TIME	*** Send Init	movb bset	LCD_CTRL,E	
Idaa	*** Wait for	movb		
*** 8 bit bus, 2 rows, 5x7 dots ldaa #\$38 jsr LCD_WRITE ;function set command ;write data to LCD *** Send Display Ctrl Command *** display on, cursor off, no blinking ldaa #\$0C jsr LCD_WRITE ;write data to LCD *** Send Clear Display Command *** clear display, cursor addr=0 ldaa #\$01 jsr LCD_WRITE ;write data to LCD movb #16T,TIME ;set delay time for 1.6ms jsr VAR_DELAY ;sub for 0.1ms delay *** Send Entry Mode Command *** increment, no display shift ldaa #\$06 jsr LCD_WRITE ;write data to LCD *** SEND MESSAGES *** Messages have address and content predefined jsr MESSAGE1 ;send Message1 ;send Message2	*** Send Init	ldaa	#\$38	
*** display on, cursor off, no blinking ldaa		, 2 rows ldaa	, 5x7 dots #\$38	
*** Send Clear Display Command *** clear display, cursor addr=0 ldaa #\$01 jsr LCD_WRITE ;write data to LCD movb #16T,TIME ;set delay time for 1.6ms jsr VAR_DELAY ;sub for 0.1ms delay *** Send Entry Mode Command *** increment, no display shift ldaa #\$06 jsr LCD_WRITE ;entry mode command jsr LCD_WRITE ;write data to LCD *** SEND MESSAGES *** Messages have address and content predefined jsr MESSAGE1 ;send Message1 jsr MESSAGE2 ;send Message2		n, curso: ldaa	r off, no blinking #\$0C	display ctrl command
ldaa #\$01 ;clear display command jsr LCD_WRITE ;write data to LCD movb #16T,TIME ;set delay time for 1.6ms jsr VAR_DELAY ;sub for 0.1ms delay *** Send Entry Mode Command *** increment, no display shift ldaa #\$06 ;entry mode command jsr LCD_WRITE ;write data to LCD *** SEND MESSAGES *** Messages have address and content predefined jsr MESSAGE1 ;send Message1 jsr MESSAGE2 ;send Message2	*** Send Clea			/WIIce data to heb
jsr LCD_WRITE ;write data to LCD movb #16T,TIME ;set delay time for 1.6ms jsr VAR_DELAY ;sub for 0.1ms delay *** Send Entry Mode Command *** increment, no display shift	*** clear dis			dianlar assumed
*** increment, no display shift ldaa #\$06 ;entry mode command jsr LCD_WRITE ;write data to LCD *** SEND MESSAGES *** Messages have address and content predefined jsr MESSAGE1 ;send Message1 jsr MESSAGE2 ;send Message2	*** Send Entr	jsr movb jsr	LCD_WRITE #16T,TIME VAR_DELAY	<pre>;write data to LCD ;set delay time for 1.6ms</pre>
*** SEND MESSAGES *** Messages have address and content predefined jsr MESSAGE1 ;send Message1 jsr MESSAGE2 ;send Message2	*** increment	, no dis ldaa	olay shift #\$06	
*** Messages have address and content predefined jsr MESSAGE1 ;send Message1 jsr MESSAGE2 ;send Message2		lar	TCD_MVIIE	WIICE data to her
<pre>jsr MESSAGE1</pre>			wood and dontont middless	
DUMMY bra DUMMY ;done with example	^^^ Messages	jsr	MESSAGE1	
	DUMMY	bra	DUMMY	;done with example

MOTOROLA 19

AN1774

```
*** Routine creates a delay according to the formula
*** TIME*~100\mus using an 8MHz internal bus
*** Cycle count per instruction shown
VAR DELAY
           ldab
                  #199T
                                            ;1
                                            ;1
L1
           nop
                                           ;3
           dbne
                  B,L1
           dec
                                           ; 4
                  TIME
                  VAR_DELAY
                                           ;3
           bne
           rts
                                           ;5
*** Routine sends LCD Data
LCD WRITEstaa LCD DATA
                  LCD CTRL, E
                                           ; clock in data
           bset
           bclr
                  LCD_CTRL, E
           ldaa
                  #107T
                                           ;40µs delay for LCD
           dbne
L2
                  A,L2
           rts
*** Routine sends LCD Address
                                           ;LCD in command mode
LCD_ADDR
           bclr
                  LCD_CTRL,RS
           staa
                  LCD DATA
           bset
                  LCD CTRL, E
                                           ; clock in data
                  LCD_CTRL, E
           bclr
           ldaa
                  #107T
                                           ;40µs delay for LCD
           dbne
                  A,L4
           bset
                  LCD CTRL, RS
                                           ;LCD in data mode
           rts
*** Message Routines
                                           ; addr = $04
MESSAGE1
           ldaa
                   #$84
           jsr
                  LCD_ADDR
                                           ;send addr to LCD
           ldx
                  #0
           ldaa
L3
                  MSG1,X
                                           ;load AccA w/char from msg
                                           ;end of msg?
           beq
                  OUTMSG1
                                           ;write data to LCD
            jsr
                  LCD_WRITE
                                           ;increment X
           inx
                                           ;loop to finish msg
           bra
                  T.3
OUTMSG1
           rts
           ldaa
                   #$C4
                                           ; addr = $44
MESSAGE 2
            jsr
                  LCD_ADDR
                                           ;send addr to LCD
           ldaa
                  MSG2,X
                                           ;load AccA w/char from msg
                   OUTMSG2
                                           ;end of msq?
           beq
                  LCD WRITE
                                            ;write data to LCD
           jsr
                                           ;increment X
           inx
           bra
                  L5
                                           ;loop to finish msg
OUTMSG2
           rts
ORG
                  MSG_STORAGE
MSG1
           db
                   'Motorola'
           db
                   'HC12 MCU'
MSG2
           db
           db
ORG
                  RESET
           DW
                  START
```

20 MOTOROLA

AN1774

4-Bit Bus Code

```
****************************
* File name: H12 LCD4.ASM
* Example Code for LCD Module (DMC16207) using 4-bit bus
   interfacing with the MC68HC912B32
* Ver: 1.0
* Date: September 6, 1998
* Author: Mark Glenewinkel
       Motorola Field Applications
* Assembler: P&E CASM12W ver 3.08
* For code explanation and flow charts, please consult Motorola Application Note
    "Interfacing the MC68HC912B32 to an LCD Module" Literature # AN1774/D
* Note: Code originates in RAM instead of FLASH
*************************
*** SYSTEM DEFINITIONS AND EQUATES ******************************
*** Internal Register Definitions
PORTA
       EQU
                 $00
                                         ;LCD data bus
          EQU
                 $01
                                         ;LCD control signals
PORTB
                                         ;data direction for PortA
DDRA
          EQU
                 $02
                                         ;data direction for PortB
DDRB
          EQU
                 $03
*** Application Specific Definitions
LCD_DATA
       EQU
              $00
                                         ; PORTA
LCD CTRL
          EQU
                 $01
                                         ; PORTB
                эо.
1Т
4Т
                                         ; PORTB, bit 0
          EQU
RW
          EOU
                                         ; PORTB, bit 2
RS
          EQU
                 2T
                                         ; PORTB, bit 1
*** Memory Definitions
RAM_START EQU $0800
                                         ;start of RAM mem
              $0BF0
$0B00
RAM_VAR
                                         ;start of RAM variables
          EQU
MSG_STORAGE EQU
                                         ;start of message block
*** Vectors
RESET
         EQU
                SFFFE
                                         ; vector for reset
ORG
                 RAM_VAR
          DB
                 1
                                         ;used for delay time
ORG RAM_START
                                         ;start at beginning of RAM
*** Initialize the Stack pointer
           lds
              #$0BFF
                                         ; init SP, top of RAM
*** Initialize Ports
START
               LCD_CTRL
                                         ;clear LCD CTRL
           clr
                LCD_DATA
           clr
                                         ;clear LCD_DATA
               #$FF,DDRA
           movb
                                         ;PortA output
           movb #$FF,DDRB
                                         ;PortB output
AN1774
```

```
*** INITIALIZE THE LCD
*** Wait for 15ms
                                                    ;set delay time
             movb
                      #150T, TIME
              jsr
                      VAR_DELAY
                                                    ; sub for 0.1ms delay
*** Send Init Command
                                                    ;LCD init command
              movb
                      #$30,LCD_DATA
              bset
                      LCD_CTRL, E
                                                    ; clock in data
              bclr
                      LCD CTRL, E
*** Wait for 4.1ms
             movb
                      #41T,TIME
                                                    ;set delay time
                                                    ; sub for 0.1ms delay
              jsr
                      VAR_DELAY
*** Send Init Command
              movb
                      #$30,LCD DATA
                                                    ;LCD init command
              bset
                                                    ; clock in data
                      LCD CTRL, E
             bclr
                     LCD_CTRL, E
*** Wait for 100 us
                      #1T,TIME
                                                    ;set delay time
             movb
                                                    ; sub for 0.1ms delay
              jsr
                      VAR_DELAY
*** Send Init Command
                      #$30
                                                    ;LCD init command
              ldaa
              jsr
                      LCD_WRITE
                                                    ;write data to LCD
*** Send Function Set Command
*** 4 bit bus, 2 rows, 5x7 dots
              ldaa
                      #$20
                                                    ;function set command
              jsr
                                                    ;write data to LCD
                      LCD WRITE
              ldaa
                     #$20
                                                    ;function set command
                                                    ;write data to LCD
              jsr
                     LCD_WRITE
                                                    ;function set command
              ldaa
                      #$80
              jsr
                      LCD_WRITE
                                                    ;write data to LCD
*** Send Display Ctrl Command
*** display on, cursor off, no blinking
              ldaa
                      #$00
                                                    ;function set command
                      LCD_WRITE
              jsr
                                                    ;write data to LCD
              ldaa
                      #$C0
                                                    ;display ctrl command
                     LCD_WRITE
                                                    ;write data to LCD
              jsr
*** Send Clear Display Command
*** clear display, cursor addr=0
              ldaa
                     #$00
                                                    ; clear display command
                                                    ;write data to LCD
              isr
                      LCD_WRITE
                                                    ;set delay time for 1.6ms
              movb
                      #16T,TIME
              jsr
                      VAR DELAY
                                                    ; sub for 0.1ms delay
              ldaa
                      #$10
                                                    ; clear display command
                      LCD WRITE
                                                    ;write data to LCD
              jsr
                      #16T,TIME
                                                    ;set delay time for 1.6ms
              movb
                                                    ; sub for 0.1ms delay
              jsr
                      VAR_DELAY
```

```
*** Send Entry Mode Command
*** increment, no display shift
                    #$00
             ldaa
                                                ;entry mode command
             jsr
                    LCD_WRITE
                                                 ;write data to LCD
             ldaa
                     #$60
                                                 ;entry mode command
                    LCD_WRITE
                                                 ;write data to LCD
             jsr
*** SEND MESSAGES
*** Messages have address and content predefined
                     MESSAGE1
             jsr
                                                 ; send Message1
                     MESSAGE2
                                                 ; send Message2
             jsr
DUMMY
            bra
                    DUMMY
                                                ; done with example
*** Routine creates a delay according to the formula
*** TIME*~100\mus using an 8MHz internal bus
*** Cycle count per instruction shown
            ldab
VAR_DELAY
                    #199T
                                                ;1
L1
                                                ;1
            nop
             dbne
                                                ;3
                    B,L1
             dec
                    TIME
                                                ; 4
                    VAR_DELAY
                                                ;3
             bne
             rts
                                                ;5
*** Routine sends LCD Data
LCD WRITE
            staa LCD DATA
            bset
                   LCD CTRL, E
                                                ;clock in data
             bclr
                   LCD CTRL, E
             ldaa
                                                ;40µs delay for LCD
                    #107T
L2
            dbne
                   A,L2
                                                ; 3
            rts
*** Routine sends LCD Address
                                                ;LCD in command mode
LCD_ADDR
            bclr LCD CTRL,RS
             staa
                    LCD_DATA
            bset
                    LCD_CTRL, E
                                                ;clock in data
            bclr
                    LCD_CTRL,E
             ldaa
                    #107T
                                                ;40\mus delay for LCD
             dbne
L4
                                                ;3
                     A,L4
                    LCD_CTRL,RS
             bset
                                                 ;LCD in data mode
             rts
*** Message Routines
                                                ; addr = $04 MSB
MESSAGE1
             ldaa
                     #$80
                                                ;send addr to LCD
             jsr
                     LCD ADDR
                                                ; addr = $04 LSB
             ldaa
                     #$40
             jsr
                     LCD ADDR
                                                ; send addr to LCD
             ldx
                     #0
L3
             ldaa
                     MSG1,X
                                                ;load AccA w/char from msq
                                                ;end of msq?
             beq
                     OUTMSG1
             jsr
                                                ;write data to LCD
                     LCD_WRITE
                                                ;load AccA w/char from msq
             ldaa
                    MSG1,X
                                                ;shift LSB to MSB
             asla
             asla
             asla
             asla
```

MOTOROLA 23

AN1774

OUTMSG1	jsr inx bra rts	LCD_WRITE	<pre>;write data to LCD ;increment X ;loop to finish msg</pre>
MESSAGE2	ldaa jsr ldaa jsr ldx	#\$C0 LCD_ADDR #\$40 LCD_ADDR #0	<pre>;addr = \$44 MSB ;send addr to LCD ;addr = \$44 LSB ;send addr to LCD</pre>
L5	ldaa beq jsr ldaa asla asla asla asla	MSG2,X OUTMSG2 LCD_WRITE MSG2,X	;load AccA w/char from msg;end of msg?;write data to LCD;load AccA w/char from msg;shift LSB to MSB
	jsr inx	LCD_WRITE	<pre>;write data to LCD ;increment X</pre>
OUTMSG2	bra rts	L5	;loop to finish msg
*** MESSAGE S		*******	*********
	ORG	MSG_STORAGE	
MSG1	db	'Motorola'	
MSG2	db db db	0 'HC12 MCU' 0	
*** VECTOR TA	ABLE **** ORG DW	************* RESET START	***********

Motorola reserves the right to make changes without further notice to any products herein. Motorola makes no warranty, representation or guarantee regarding the suitability of its products for any particular purpose, nor does Motorola assume any liability arising out of the application or use of any product or circuit, and specifically disclaims any and all liability, including without limitation consequential or incidental damages. "Typical" parameters which may be provided in Motorola data sheets and/or specifications can and to vary in different applications and actual performance may vary over time. All operating parameters, including "Typicals" must be validated for each customer application by customer's technical experts. Motorola does not convey any license under its patent rights nor the rights of others. Motorola products are not designed, intended, or authorized for use as components in systems intended for surgical implant into the body, or other applications intended to support or sustain life, or for any other application in which the failure of the Motorola product could create a situation where personal injury or death may occur. Should Buyer purchase or use Motorola products for any such unintended or unauthorized application, Buyer shall indemnify and hold Motorola and its officers, employees, subsidiaries, affiliates, and distributors harmless against all claims, costs, damages, and expenses, and reasonable attorney fees arising out of, directly or indirectly, any claim of personal injury or death associated with such unintended or unauthorized use, even if such claim alleges that Motorola was negligent regarding the design or manufacture of the part. Motorola and (A) are registered trademarks of Motorola, Inc. Motorola, Inc. is an Equal Opportunity/Affirmative Action Employer.

How to reach us:

USA/EUROPE/Locations Not Listed: Motorola Literature Distribution, P.O. Box 5405, Denver, Colorado 80217.

1-800-441-2447 or 1-303-675-2140. Customer Focus Center, 1-800-521-6274

JAPAN: Motorola Japan Ltd.: SPD, Strategic Planning Office, 141, 4-32-1 Nishi-Gotanda, Shinagawa-Ku, Tokyo, Japan, 03-5487-8488 ASIA/PACIFIC: Motorola Semiconductors H.K. Ltd., Silicon Harbour Centre, 2 Dai King Street, Tai Po Industrial Estate,

Tai Po, New Territories, Hong Kong, 852-26629298

Mfax™, Motorola Fax Back System: RMFAX0@email.sps.mot.com; http://sps.motorola.com/mfax/;

TOUCHTONE, 1-602-244-6609; US and Canada ONLY, 1-800-774-1848

HOME PAGE: http://motorola.com/sps/

Mfax is a trademark of Motorola, Inc.

© Motorola, Inc., 1999

